Appendices:  SAS codes used to conduct the various tests
Appendix 1. Performing non-parametric tests (NPARTs)
Proc NPAR1WAY Data = Dorm WILCOXON;
Title 'The nonparametric procedure producing Wilcoxon’s exact test';
Class variety;
Var Germ; /*Germination proportion*/
EXACT WILCOXON/MC;/*Exact P values based on Monte Carlo simulation */
Run;

PROC RANK DATA=Dorm OUT=Dormrank;
VAR Germ;
run;

PROC GLM DATA=Dormrank; /*Performing multiple Comparisons on Wilcoxon ranks*/
Title 'Treatment vs control: Bonferroni, Sidak and Dunnett adjustmen';
CLASS variety;
MODEL Germ = variety/ss3;
Lsmeans variety/adjust=BON pdiff=control ('V5a') cl  ;
Lsmeans variety/adjust=SIDAK pdiff=control ('V5a') cl;
Lsmeans variety/adjust=DUNNETT pdiff=control ('V5a') cl;
run;

PROC GLM DATA=Dormrank;
CLASS variety;
MODEL Germ = variety/ss3;
Means variety/LSD SNK DUNCAN TUKEY SCHEFFE BON SIDAK;
run;

Appendix 2. Performing ANOVA using the general linear model (GLM)
Proc GLM Data = Dorm;
Title 'The GLM procedure for ANOVA';
Class Rep Cont variety;
Model Germ = Rep variety/ss3;
Means variety/ LSD SNK DUNCAN TUKEY SCHEFFE WALLER REGWQ;
Output Out = Dorm1 p = Pred r = Res; /*Res are residuals saved in the file Dorm1 and will be used in test of normality as follows*/
Run;

Proc Univariate Normal Data = Dorm1;
Title 'Test for normality of residuals';
Var Res;
Run;

Proc GLM data = Dorm1;
Title 'Tukey test for additivity';
Class Rep Cont variety;
Model Germ = Rep variety Pred*Pred/ss3;/*Pred*Pred provides a test of additivity*/;

Proc GLM Data = Dorm;
Title 'Treatment vs control: Bonferroni, Sidak and Dunnett adjustmen';
Class variety;
Model Germ = variety;
Lsmeans variety/adjust=BON pdiff=control ('V5a')cl;
Lsmeans variety/adjust=SIDAK pdiff=control ('V5a') cl;
Lsmeans variety/adjust=DUNNETT pdiff=control ('V5a') cl;
Run;

Proc GLM Data = Dorm;
Title 'The following code performs homogeneity of variance tests';
Class variety;
Model Germ = variety;
Means variety/hovtest=LEVENE; /*Gives Leven’s test. Replace Levene with BARTLETT, BF or OBRIEN for the corresponding test*/
Run;

Appendix 3. Performing linear mixed model analysis
proc mixed data=Dorm CL CovTest;
Title 'Linear mixed model analysis of variance';
Class Rep Cont variety;
model Germ = Rep variety/ddfm=kr;
Random Cont variety*Rep;
lsmeans variety/diff adjust=TUKEY; /*Gives Tukey-Kramer adjusted P values*/
lsmeans variety/diff adjust=SIDAK; /*Gives Sidak adjusted P values*/
lsmeans variety/diff adjust=BON; /*Gives Bonferroni adjusted P values*/
run;
Appendix 4. Performing logistic regression using the generalized linear model (GLM)
proc genmod data=Dorm;
Title 'Logistic regression model';
class Rep Cont variety; 
model n/T = Rep variety/dist=binomial link=logit scale=deviance type3;
/*n=number of germinated seeds; T = total seeds*/
lsmeans variety/ diff;
run;
Appendix 5. Performing generalized linear mixed models (GLMMs)
ods graphics on;
proc glimmix data=seeds plots=all;  
Title 'Random effect model using generalized linear mixed model';
Class Rep Cont variety;
model n/T = Rep variety/dist=binomial link=logit ddfm=kr; 
random Cont(variety *Rep) /type=vc;
lsmeans variety/ diff adjust=TUKEY;
run;
ods graphics off;
ods pdf;


